
Excel Kantor + AI
Rumus yang paling dipakai admin, HR, keuangan, dan guru, lengkap
dengan 24 template siap pakai dan cara menyuruh ChatGPT/Gemini
membuat rumus yang benar.

fx =DATEDIF(F5;Pengaturan!$B$4;"Y")&" th "&DATEDIF(F5;Pengaturan!$B$4;"YM")&" bln"

A B F K O

4 NIK Nama Tgl Masuk Masa Kerja Status Kontrak

5 K-0001 Rina Utami 15/03/2018 8 th 7 bln

6 K-0002 Galih Nugraha 02/01/2025 1 th 9 bln Segera berakhir

7 K-0003 Sari Wijaya 01/11/2016 9 th 11 bln

PROMPT

Excel 2016, pemisah titik koma. Tanggal masuk di kolom F mulai
baris 5. Tampilkan masa kerja "8 th 7 bln"; kalau kosong, hasil
kosong.

XLOOKUP & INDEX-MATCH SUMIFS PivotTable Absensi & Lembur PPh 21 TER Terbilang

Invoice Rapor Dashboard Prompt AI

Aphriell Art x Ares Studio

© 2026 Aphriell Art x Ares Studio. Hak cipta dilindungi undang-undang. Contoh gratis ini boleh dibagikan utuh tanpa

Merek dagang: Microsoft, Excel, Word, dan Copilot adalah merek dagang Microsoft Corporation. Google, Google Sheets, dan
Gemini adalah merek dagang Google LLC. ChatGPT adalah merek dagang OpenAI. Claude adalah merek dagang Anthropic.
Ebook ini tidak berafiliasi dengan maupun disponsori oleh pihak-pihak tersebut.

Penafian: Contoh perhitungan pajak, BPJS, dan ketenagakerjaan disusun berdasarkan aturan yang berlaku per November 2026
dan bisa berubah; ini bukan nasihat hukum atau pajak. Seluruh nama, NIK, perusahaan, dan angka dalam contoh adalah fiktif.

Kata Pengantar

Hampir semua pekerjaan kantor di Indonesia lewat Excel. Absensi, slip gaji, stok gudang, daftar harga, rekap
penjualan, nilai siswa, kuitansi. Masalahnya, kebanyakan dari kita belajar Excel sambil jalan: menyalin rumus
dari grup WhatsApp, mencari di internet, dan mencoba sampai angkanya "kelihatan benar".

Cara itu bisa berhasil, tapi mahal. Rumus dari internet error di laptop kantor karena pemisahnya beda.
Laporan bulanan dikerjakan ulang dari nol setiap bulan. Satu sel yang tertimpa angka manual membuat total
salah tanpa ada yang sadar sampai ditanya atasan.

Ebook ini ditulis untuk mengubah itu, dengan tiga prinsip.

Pertama, disusun per pekerjaan, bukan per fungsi. Anda tidak akan menemukan bab "Fungsi DATEDIF".
Yang ada adalah bab "Umur, Masa Kerja, dan Tanggal Penting", karena itulah yang Anda cari saat diminta
HRD menghitung masa kerja 80 karyawan. Rumusnya dijelaskan di tempat ia dibutuhkan.

Kedua, setiap bab punya file kerja. Paket ini berisi 24 template Excel siap pakai. Buka, ganti data contoh
dengan data Anda, selesai. Kalau kasus Anda sedikit berbeda, bab yang membahasnya menjelaskan cara
menyesuaikannya.

Ketiga, AI dipakai sebagai asisten, bukan pengganti. ChatGPT, Gemini, dan Copilot bisa membuat rumus
dalam hitungan detik. Mereka juga bisa salah dengan sangat meyakinkan. Bab 2 mengajarkan cara meminta
rumus ke AI dan cara memeriksanya, keterampilan yang menurut kami jauh lebih berharga daripada
menghafal ratusan fungsi.

Untuk siapa ebook ini
Staf admin dan umum yang mengurus data barang, surat, dan laporan.

HR dan payroll yang menghitung absensi, lembur, cuti, dan gaji.
Staf keuangan, kasir, dan pemilik UMKM yang membuat invoice, kuitansi, kas kecil, dan laba rugi.

Guru dan staf TU yang merekap nilai dan absensi siswa.
Pencari kerja yang akan menghadapi tes Excel. Bab 23 berisi 30 soal latihan yang dicek otomatis.

Anda tidak perlu mahir. Kalau Anda bisa mengetik di sel dan tahu bahwa rumus diawali tanda = , Anda siap
mulai.



Excel Kantor + AI Kata Pengantar

© 2026 Aphriell Art x Ares Studio 5

Versi Excel yang dipakai
Contoh di ebook ini diuji di Microsoft Excel 365. Rumus utama di setiap template sengaja dibatasi pada
fungsi yang ada sejak Excel 2010, sehingga tetap berjalan di Excel 2016 dan 2019 yang masih banyak dipakai
kantor, serta di Google Sheets. Fungsi yang hanya ada di versi baru, seperti XLOOKUP dan LAMBDA, selalu
diberi label "Excel 365" dan disertai pengganti untuk versi lama.

Semua nama, angka, NIK, rekening, dan perusahaan di ebook dan template ini fiktif.

Selamat bekerja lebih cepat, dan pulang lebih awal.

Aphriell Art x Ares Studio November 2026

Cara Membaca Ebook Ini

Bab 1–3 adalah fondasi. Bacalah lebih dulu, terutama Bab 1, karena di sana dijelaskan kenapa rumus yang
benar bisa error di komputer Anda. Setelah itu, lompatlah ke bab yang sesuai pekerjaan Anda. Setiap bab
pekerjaan berdiri sendiri.

Pola setiap bab pekerjaan
1. Kasus: cerita singkat dari pekerjaan sehari-hari.
2. Hasil akhir: seperti apa tabel yang akan kita buat.

3. Langkah dan rumus: rumus inti, dibedah per bagian.
4. Jebakan umum: kesalahan yang paling sering terjadi dan cara menghindarinya.

5. Excel lama dan Google Sheets: pengganti bila fungsi tertentu tidak tersedia.
6. Prompt AI: kalimat siap salin untuk menyesuaikan rumus dengan kasus Anda.

7. Template: nama file di paket ZIP yang memakai rumus bab tersebut.

Tanda yang dipakai
Tabel 1. Penanda di dalam ebook

Penanda Artinya

Kotak hijau Tips atau ringkasan penting

Kotak jingga Jebakan: sumber kesalahan yang sering terjadi

Teks seperti ini Rumus, nama fungsi, nama sheet, atau isi sel

Sel kuning di template Boleh diisi atau diganti

Sel abu-abu di template Rumus, jangan diketik ulang

Tentang pemisah rumus: titik koma atau koma?
Di ebook ini rumus ditulis dengan titik koma ( ; ) sebagai pemisah argumen, karena itulah yang muncul di
Excel dengan pengaturan wilayah Indonesia, pengaturan yang dipakai sebagian besar komputer kantor di
sini:

=IF(B2>=75;"Tuntas";"Remedial")

Kalau Excel Anda memakai koma, ganti setiap ;  dengan , . Rumusnya sama persis. Bab 1.2 menjelaskan
cara mengecek pengaturan Anda.

Memakai template
Template ada di folder Template/  dalam file ZIP. Setiap file punya sheet Petunjuk di depan yang
menjelaskan isi dan cara pakainya. Sebelum mengisi data asli:

1. Simpan salinan file (File → Save As) dengan nama baru. Template asli tetap utuh.

2. Hapus data contoh di sel kuning, jangan menghapus barisnya.
3. Isi data Anda di sel kuning. Sel abu-abu akan menghitung sendiri.



Excel Kantor + AI Sebelum Menulis Rumus

© 2026 Aphriell Art x Ares Studio 8

B A B  1

Sebelum Menulis Rumus

Senin pagi, Dimas, staf admin baru, menemukan rumus yang ia butuhkan di sebuah artikel. Ia menyalinnya ke
Excel kantor, menekan Enter, dan muncul kotak peringatan: "There's a problem with this formula." Di laptop
pribadinya, rumus yang sama langsung jalan.

Rumusnya tidak salah. Yang berbeda adalah pengaturan komputernya. Bab ini membahas hal-hal seperti
itu: hal kecil yang membuat rumus benar terlihat salah, dan yang membuat rumus salah terlihat benar.
Setelah bab ini, sebagian besar error yang Anda temui akan punya penjelasan.

1.1  Versi Excel Anda menentukan fungsi yang tersedia
Excel terus menambah fungsi baru. File yang dibuat di Excel 365 bisa berisi fungsi yang belum dikenal Excel
2016. Saat dibuka di komputer lama, sel tersebut menampilkan #NAME? .

Tabel 2. Fungsi populer dan versi Excel minimal

Fungsi Tersedia sejak Dipakai untuk

SUMIFS , COUNTIFS , AVERAGEIFS , IFERROR Excel 2007 Menjumlah/menghitung bersyarat, menangani
error

NETWORKDAYS.INTL , WORKDAY.INTL , RANK.EQ Excel 2010 Hari kerja dengan pola libur bebas, peringkat

IFNA , DAYS Excel 2013 Menangani #N/A  saja, selisih hari

IFS , MAXIFS , MINIFS , TEXTJOIN , CONCAT ,
SWITCH

Excel 2019 Syarat bertingkat, nilai maks/min bersyarat, gabung
teks

XLOOKUP , FILTER , UNIQUE , SORT , SEQUENCE ,
LET

Excel 2021 /
365

Pencarian modern, hasil yang "tumpah" ke banyak
sel

LAMBDA , VSTACK , TEXTSPLIT Excel 365 /
2024

Fungsi buatan sendiri, olah larik

Cara mengecek versi Anda: File → Account → About Excel. Kalau tertulis "Microsoft 365", Anda punya fungsi
terbaru. Kalau tertulis 2016 atau 2019, fungsi di dua baris terbawah tabel tidak tersedia.

Aturan praktis. Kalau file akan dibuka orang lain, misalnya laporan ke atasan atau file bersama satu
divisi, pakai fungsi yang tersedia di versi paling lama yang dipakai di kantor Anda. Semua template di
paket ini memakai fungsi Excel 2010 untuk rumus utamanya karena alasan itu.

1.2  Titik koma atau koma: pemisah argumen
Ini penyebab nomor satu "rumus dari internet tidak jalan". Bandingkan dua rumus ini:

=IF(B2>=75;"Tuntas";"Remedial")

=IF(B2>=75,"Tuntas","Remedial")

Keduanya rumus yang sama. Yang pertama memakai titik koma, yang kedua koma. Excel hanya menerima
salah satunya, tergantung pengaturan wilayah (regional settings) di Windows:

Pengaturan Indonesia: angka desimal ditulis dengan koma ( 2,5 ), sehingga pemisah argumen rumus
harus titik koma.
Pengaturan English (United States): desimal ditulis dengan titik ( 2.5 ), sehingga pemisah argumen bisa
koma.

Cara tercepat mengecek: ketik =SUM(  di sel mana pun, lalu lihat kotak petunjuk kecil di bawahnya. Kalau
tertulis SUM(number1; [number2]; ...) , Excel Anda memakai titik koma.

Memperbaiki rumus hasil salin

Kalau rumus yang Anda salin memakai koma sedangkan Excel Anda titik koma (atau sebaliknya):

1. Tempel rumus ke Notepad atau langsung ke formula bar.

2. Ganti setiap koma pemisah argumen dengan titik koma.
3. Hati-hati dengan koma yang bukan pemisah: koma di dalam teks berkutip ( "Jakarta, Indonesia" ) dan

angka desimal. Angka 0.5  di rumus berpemisah koma harus menjadi 0,5  di rumus berpemisah titik
koma.

Kabar baiknya: pemisah hanya soal tampilan saat mengetik. File Excel menyimpan rumus dalam format
netral. File dari komputer berpengaturan Inggris tetap jalan di komputer berpengaturan Indonesia; Excel
menerjemahkan pemisahnya otomatis. Masalah hanya muncul saat Anda mengetik atau menempel
teks rumus.

Jebakan yang lebih halus: kode format di dalam TEXT

Pengaturan wilayah juga memengaruhi kode format di dalam fungsi TEXT . Rumus berikut menampilkan
angka dengan pemisah ribuan di komputer berpengaturan Inggris:

=TEXT(A2;"#,##0")

Di komputer berpengaturan Indonesia, koma di dalam kode format itu dibaca sebagai tanda desimal,
sehingga hasilnya berantakan. Hal serupa terjadi pada kode format tanggal di dalam TEXT  untuk sebagian
bahasa.

Jebakan. Hindari TEXT  dengan kode format ribuan atau tanggal di file yang dibuka banyak orang.
Untuk angka berpemisah ribuan di dalam kalimat, pakai FIXED(A2;0) , yang otomatis mengikuti
pengaturan komputer pembacanya. Untuk tampilan angka di sel, cukup atur format selnya (Ctrl+1).
Untuk tanggal dalam kalimat, lihat cara di Bab 8 dan Bab 9.

1.3  Referensi relatif, absolut, dan campuran
Saat rumus disalin ke bawah atau ke samping, alamat sel di dalamnya ikut bergeser. Itu biasanya yang kita
mau, tapi tidak selalu.

Misalkan komisi sales adalah 2,5% dan disimpan di sel F1 . Penjualan ada di kolom B mulai baris 4:

=B4*F1

Rumus di C4  benar. Tapi begitu disalin ke C5 , ia berubah menjadi =B5*F2 , padahal F2  kosong, sehingga
komisinya nol. Solusinya, kunci alamat F1  dengan tanda dolar:

=B4*$F$1

Tabel 3. Jenis referensi

Tulisan Saat disalin ke bawah Saat disalin ke kanan Dipakai untuk

F1 baris bergeser kolom bergeser Data yang ikut bergeser per baris

$F$1 tetap tetap Satu sel pengaturan: tarif, tanggal acuan, KKM

$F1 baris bergeser kolom tetap Tabel dua arah: kolom kriteria terkunci

F$1 baris tetap kolom bergeser Tabel dua arah: baris judul terkunci

Anda tidak perlu mengetik tanda dolar. Klik alamat sel di formula bar lalu tekan F4 berulang kali untuk
berganti di antara empat jenis di atas.

Referensi campuran dalam praktik

Referensi campuran paling berguna untuk tabel rekap dua arah, misalnya penjualan per bulan (baris) per
cabang (kolom). Anggap sheet Data  punya kolom Cabang di C, Total di I, dan Bulan di K (berisi 2026-01 ,
2026-02 , dan seterusnya). Di sheet rekap, kode bulan ditulis di kolom A dan nama cabang di baris 5:

=SUMIFS(Data!$I$5:$I$604;Data!$K$5:$K$604;$A6;Data!$C$5:$C$604;B$5)

Rentang data dikunci penuh ( $I$5:$I$604 ). Kode bulan di $A6  dikunci kolomnya karena selalu di kolom A.
Nama cabang di B$5  dikunci barisnya karena selalu di baris judul. Hasilnya, satu rumus bisa disalin ke
seluruh tabel 12 × 4 tanpa diubah. Sheet "Rekap Rumus" di template 03-Rekap-Penjualan-Pivot.xlsx

memakai pola yang sama, ditambah sel tahun yang bisa diganti.

1.4  Bentuk data yang "disukai" rumus
Rumus Excel bekerja paling baik pada data yang tersusun rapi. Sebagian besar masalah "rumus tidak mau
menghitung" sebenarnya masalah bentuk data.

1. Satu baris judul, lalu data. Jangan ada judul bertingkat, baris kosong di tengah, atau baris "subtotal"
yang diselipkan di antara data.

2. Satu sel, satu data. Jangan menulis Jakarta - 0812xxxx  dalam satu sel; pisahkan kota dan nomor HP.

3. Jangan menggabungkan sel (merge) di tabel data. Merge boleh untuk judul laporan, jangan untuk
tabel yang akan diolah.

4. Angka harus berupa angka. Angka yang tersimpan sebagai teks biasanya rata kiri dan ada segitiga hijau
di pojok sel. SUM  mengabaikannya diam-diam, sehingga total Anda kurang tanpa peringatan.

5. Tanggal harus berupa tanggal. Tanggal hasil salin dari sistem lain sering tersimpan sebagai teks. Cara
mengeceknya dibahas di Bab 9.1, dan cara membersihkannya di Bab 5.

Cara cepat mengecek angka teks. Pilih kolomnya, lalu lihat status bar di bawah layar. Kalau hanya
muncul Count  tanpa Sum , seluruh isinya teks. Kalau Sum  muncul, klik kanan status bar, aktifkan
Numerical Count, lalu bandingkan dengan Count . Kalau angkanya berbeda, sebagian sel berisi teks.

Tabel Excel (Ctrl+T): kapan dipakai

Fitur Format as Table (Ctrl+T) mengubah rentang biasa menjadi "Tabel" yang otomatis melebar saat data
ditambah, memberi warna selang-seling, dan menyalin rumus ke baris baru dengan sendirinya. Untuk data
yang Anda olah sendiri di Excel, fitur ini sangat membantu.

Template di paket ini sengaja tidak memakai Tabel. Rumusnya sudah disiapkan untuk ratusan baris dengan
pola =IF(A5="";"";...)  agar baris kosong tetap bersih. Pola ini dipilih karena tetap berfungsi saat file
dibuka di Google Sheets dan saat dipakai sebagai sumber mail merge Word, dua situasi di mana Tabel Excel
kadang berperilaku berbeda.

1.5  Membaca pesan error
Pesan error bukan musuh. Ia memberi tahu jenis masalahnya.

Tabel 4. Arti pesan error dan penyebab paling umum

Error Artinya Penyebab paling umum

#N/A Data yang dicari tidak
ditemukan

Kode/nama beda tipis (spasi berlebih, huruf O vs angka 0), atau memang tidak
ada

#VALUE! Jenis data salah Menghitung teks sebagai angka, tanggal tersimpan sebagai teks

#REF! Alamat sel tidak ada lagi Baris/kolom/sheet yang dirujuk sudah dihapus

#DIV/0! Pembagian dengan nol Penyebut kosong, misalnya rata-rata dari data yang belum diisi

#NAME? Nama tidak dikenal Salah ketik nama fungsi, teks tanpa tanda kutip, atau fungsi tidak ada di versi
Excel Anda

#NUM! Angka tidak valid Tanggal awal lebih besar dari tanggal akhir di DATEDIF , hasil terlalu besar

#SPILL! Hasil tidak bisa "tumpah" (Excel 365) Sel di bawah/kanan rumus dinamis sudah berisi sesuatu

##### Kolom terlalu sempit Bukan error. Lebarkan kolomnya. Bisa juga tanggal bernilai negatif

Hati-hati dengan IFERROR

IFERROR  menyembunyikan error dan menggantinya dengan nilai lain:

=IFERROR(VLOOKUP(A5;Harga!$A$5:$E$104;5;FALSE);0)

Rumus di atas menampilkan 0  bila kode barang tidak ditemukan. Masalahnya, 0  di laporan terlihat seperti
data sah. Salah ketik kode barang jadi tidak ketahuan, dan total penjualan diam-diam kurang.

Jebakan. Jangan menutupi error dengan angka 0 atau sel kosong di data yang dijumlahkan. Ganti
dengan pesan yang terlihat jelas, misalnya IFERROR(...;"Kode salah") , lalu beri warna merah lewat
Conditional Formatting. Template 04 (Stok Barang) memakai cara ini di kolom nama barang.

1.6  Ringkasan
Fungsi baru (XLOOKUP, FILTER, LAMBDA) tidak ada di Excel lama. Untuk file bersama, pakai fungsi versi
paling lama di kantor Anda.

Pemisah ;  atau ,  mengikuti pengaturan wilayah Windows. Rumus dari internet mungkin perlu diganti
pemisahnya.
Hindari kode format di dalam TEXT  untuk file bersama; pakai FIXED  atau format sel.

Tekan F4 untuk mengunci alamat sel. Sel pengaturan selalu $F$1 .
Rapikan bentuk data sebelum menyalahkan rumus: satu baris judul, satu sel satu data, angka benar-benar
angka.
Pesan error memberi petunjuk. Jangan buru-buru membungkamnya dengan IFERROR(...;0) .

B A B  2

Menyuruh AI Membuat Rumus (dan
Memeriksanya)

Sari, staf HRD, diminta menambahkan kolom masa kerja di data karyawan. Ia membuka ChatGPT, mengetik
"rumus excel masa kerja", dan mendapat rumus dalam tiga detik. Ia menempelnya, dan muncul pesan error.
Setelah pemisahnya diganti, rumus berjalan. Ia menyalinnya ke 80 baris dan mengirim filenya ke manajer.

Dua hari kemudian manajer bertanya kenapa ada karyawan dengan masa kerja 126 tahun.

AI tidak berbohong kepada Sari. AI menjawab pertanyaan yang ia ajukan, sebuah pertanyaan yang terlalu
pendek. Bab ini mengajarkan dua hal: cara bertanya supaya rumus dari AI langsung cocok dengan data
Anda, dan cara menguji rumus itu sebelum dipakai.

2.1  Apa yang bisa dan tidak bisa dikerjakan AI
Asisten AI seperti ChatGPT, Gemini, Copilot, dan Claude sangat bagus untuk:

membuat rumus dari penjelasan bahasa sehari-hari;
menjelaskan rumus panjang yang Anda warisi dari pendahulu, bagian per bagian;

mencari tahu penyebab error bila Anda memberi rumus dan pesan error-nya;
mengusulkan cara lain: "ada cara yang lebih sederhana?".

Tetapi AI tidak melihat file Anda (kecuali Anda memakai fitur yang terhubung langsung ke Excel, dibahas di
2.8). Ia tidak tahu kolom mana berisi apa, baris berapa data dimulai, versi Excel Anda, atau pemisah yang
Anda pakai. Ia akan menebak, dan tebakannya terdengar sangat yakin.

Prinsip bab ini. Semakin sedikit AI perlu menebak, semakin benar rumusnya. Tugas Anda adalah
mengganti tebakan dengan informasi.

2.2  Pola prompt lima bagian
Prompt yang baik untuk rumus Excel memuat lima hal.

Tabel 5. Lima bagian prompt rumus Excel

Bagian Isi Contoh

1. Pekerjaan Apa yang sedang Anda kerjakan "Saya staf HRD, membuat data karyawan."

2. Letak data Nama sheet, kolom, baris awal,
contoh isi

"Sheet Karyawan. Tanggal masuk di kolom F, mulai baris 5.
Contoh: 12/03/2018."

3. Hasil yang
diinginkan

Kolom tujuan dan contoh hasil "Di kolom K tampil masa kerja, contoh: 8 th 7 bln."

4. Batasan Versi Excel, pemisah, kondisi
khusus

"Excel 2016, pemisah titik koma. Kalau tanggal masuk kosong,
hasil kosong."

5. Permintaan
penjelasan

Minta rumus dijelaskan "Jelaskan tiap bagian rumusnya dengan singkat."

Bagian 3 dan 4 paling sering dilupakan, padahal keduanya yang paling menentukan.

2.3  Prompt buruk dan prompt baik
Prompt Sari:

rumus excel masa kerja

Jawaban yang umum muncul untuk prompt sependek ini kira-kira seperti berikut (bentuk persisnya bisa
berbeda):

=DATEDIF(C2,TODAY(),"y")&" tahun "&DATEDIF(C2,TODAY(),"ym")&" bulan"

Rumus ini benar untuk data contoh di kepala AI, tapi di file Sari ada empat masalah:

1. Pemisahnya koma, sedangkan Excel Sari titik koma, sehingga muncul pesan error.
2. Tanggal masuk dianggap di C2 , padahal di file Sari ada di kolom F mulai baris 5.

3. Tidak ada penanganan sel kosong. Untuk baris karyawan yang tanggal masuknya belum diisi, DATEDIF
menghitung dari "tanggal 0", yaitu awal tahun 1900, sehingga muncul 126 tahun.

4. Memakai TODAY() , sehingga angka berubah setiap hari. Untuk laporan per akhir bulan, ini bisa jadi
masalah (Bab 9.2).

Prompt yang lebih baik:

Saya staf HRD dan membuat data karyawan di Excel 2016 berbahasa Indonesia (pemisah rumus titik
koma). Di sheet "Karyawan", tanggal masuk ada di kolom F mulai baris 5, formatnya tanggal asli (contoh
12/03/2018). Tanggal acuan laporan ada di sheet "Pengaturan" sel B4. Saya ingin di kolom K tampil
masa kerja dalam format "8 th 7 bln". Kalau tanggal masuk kosong, kolom K kosong. Kalau tanggal
masuk lebih besar dari tanggal acuan, tulis "Belum masuk". Berikan rumus untuk sel K5 yang bisa saya
salin ke bawah, lalu jelaskan tiap bagiannya.

Jawaban untuk prompt seperti ini biasanya sudah siap pakai:

=IF(F5="";"";IF(F5>Pengaturan!$B$4;"Belum masuk";DATEDIF(F5;Pengaturan!$B$4;"Y")&" th

"&DATEDIF(F5;Pengaturan!$B$4;"YM")&" bln"))



Excel Kantor + AI Menyuruh AI Membuat Rumus (dan Memeriksanya)

© 2026 Aphriell Art x Ares Studio 15

Ini persis rumus yang dipakai di template 07-Data-Karyawan.xlsx . Perhatikan $B$4  dikunci (Bab 1.3),
sehingga rumus aman disalin ke bawah.

2.4  Lima cara AI salah membuat rumus
Kesalahan pada rumus buatan AI hampir selalu jatuh ke salah satu dari lima kelompok ini. Kenali polanya, dan
Anda akan menemukannya dalam hitungan detik.

1. Pemisah dan format angka

AI dilatih terutama dengan contoh berbahasa Inggris, sehingga ia cenderung memakai koma sebagai
pemisah dan titik sebagai desimal ( 0.5 ). Sebutkan pemisah Anda di prompt, atau ganti sendiri seperti di Bab
1.2.

2. Fungsi yang tidak ada di Excel Anda

Tanpa keterangan versi, AI sering memakai XLOOKUP , FILTER , LET , atau TEXTJOIN . Di Excel 2016 hasilnya
#NAME? . Sebutkan versi Excel Anda, atau minta: "pakai fungsi yang ada di Excel 2010".

3. Alamat sel dan rentang yang salah

AI tidak tahu baris judul Anda di baris 4 dan data mulai baris 5. Ia biasanya menebak A2 . Rentang seperti
A2:A100  juga sering terlalu pendek untuk data Anda. Beri tahu letak persisnya.

4. Kasus pinggir tidak ditangani

Ini kesalahan yang paling berbahaya, karena tidak memunculkan error. Rumus berjalan, angkanya keluar,
tapi salah. Contohnya:

sel kosong dihitung sebagai 0 (masa kerja 126 tahun, umur 126 tahun);
nilai tepat di batas: apakah nilai 75 itu "Tuntas" kalau KKM-nya 75? Tergantung >  atau >= ;

pembagian dengan nol saat data belum lengkap;
teks berbeda tipis: "Produksi "  (ada spasi) dianggap berbeda dari "Produksi" .

5. Fungsi karangan

Kadang AI menyebut fungsi yang tidak ada di Excel bawaan, dengan sangat meyakinkan. Contoh klasik di
Indonesia: =TERBILANG(A1) . Fungsi ini hanya ada jika komputer memasang add-in terbilang.xla  atau file
punya makro sendiri. Di komputer lain, hasilnya #NAME? . Kalau AI memberi fungsi yang belum pernah Anda
dengar, tanyakan balik: "Apakah fungsi ini bawaan Excel 2016 tanpa add-in?". Bab 13 menunjukkan cara
membuat terbilang tanpa add-in.

2.5  Menguji rumus sebelum dipakai
Jangan menilai rumus dari satu baris yang "kelihatan benar". Siapkan data uji kecil yang sengaja memancing
kesalahan, lalu bandingkan hasil rumus dengan hitungan manual Anda.

Tabel 6. Contoh data uji untuk rumus masa kerja (tanggal acuan 31/10/2026)

Kasus Tanggal masuk Hasil yang benar Yang diuji

Normal 15/03/2018 8 th 7 bln Rumus dasar

Tepat ulang tahun kerja 31/10/2016 10 th 0 bln Batas tahun

Sehari sebelum ulang tahun kerja 01/11/2016 9 th 11 bln Tidak dibulatkan ke atas

Baru masuk bulan ini 01/10/2026 0 th 0 bln Hasil nol

Belum masuk 01/12/2026 Belum masuk Tanggal setelah acuan

Kosong (kosong) (kosong) Sel kosong

Teks, bukan tanggal "12-03-2018" Error, perlu dibersihkan Bentuk data

Enam sampai delapan baris uji seperti ini cukup untuk menangkap hampir semua kesalahan di bagian
sebelumnya. Simpan baris uji di sheet terpisah. Kalau suatu hari rumusnya Anda ubah, uji ulang dengan data
yang sama.

Kebiasaan yang menyelamatkan. Untuk setiap rumus baru, hitung satu baris secara manual (pakai
kalkulator atau kalender) dan cocokkan. Kalau satu baris manual cocok dan semua baris uji lolos, Anda
bisa menyalinnya ke ribuan baris dengan tenang.

2.6  Meminta AI memperbaiki rumus
Kalau rumus error atau hasilnya aneh, jangan hanya menulis "rumusnya error". Beri AI tiga hal: rumusnya,
pesan error atau hasil yang salah, dan isi sel yang dirujuk.

Rumus ini di sel K12 menghasilkan #VALUE!: =DATEDIF(F12;Pengaturan!$B$4;"Y")  Isi F12 terlihat
seperti tanggal "12-03-2018" tapi rata kiri. Pengaturan!B4 berisi 31/10/2026. Apa penyebabnya dan
bagaimana memperbaikinya di Excel 2016?

Dengan informasi itu, AI biasanya langsung menebak dengan benar: F12 berisi teks yang mirip tanggal.
Perbaikannya membersihkan datanya (Bab 5), bukan membuat rumus makin rumit.

Permintaan lain yang sangat berguna:

"Jelaskan rumus ini bagian per bagian seperti untuk pemula: [tempel rumus]". Cocok untuk rumus warisan
yang tidak ada dokumentasinya.
"Rumus ini terlalu panjang. Ada cara yang lebih sederhana di Excel 2016?"

"Buatkan 6 baris data uji untuk rumus ini, termasuk kasus pinggirnya."

2.7  Privasi: jangan tempel data asli
Saat menempel isi tabel ke layanan AI, data itu dikirim ke server pihak lain. Untuk data pribadi seperti nama
lengkap, NIK, nomor HP, alamat, gaji, rekening, atau nilai siswa, ini menyangkut UU No. 27 Tahun 2022
tentang Pelindungan Data Pribadi dan hampir pasti juga kebijakan kantor Anda.

Untungnya, AI tidak butuh data asli untuk membuat rumus. Ia hanya butuh bentuk datanya.

Tabel 7. Data asli vs data contoh untuk prompt

Jangan ditempel Cukup tulis

3273015505920002 · Sari Wijaya · 8.750.000 "Kolom C berisi NIK 16 digit, kolom D nama, kolom G gaji (angka)."

80 baris data karyawan 2–3 baris contoh dengan nama dan angka karangan

Screenshot tabel gaji Deskripsi kolom dan satu contoh baris fiktif

Aturan praktis. Kalau Anda ragu apakah suatu data boleh ditempel ke AI, anggap tidak boleh. Ganti
dengan data karangan yang bentuknya sama. Rumusnya akan tetap benar.

Beberapa kantor menyediakan AI versi perusahaan, misalnya Copilot di Microsoft 365 berlisensi kantor, yang
perjanjian datanya berbeda dengan versi gratis. Tanyakan ke tim IT Anda mana yang boleh dipakai untuk data
kantor.

2.8  AI yang terhubung langsung ke spreadsheet
Per November 2026, beberapa asisten AI bisa bekerja langsung di dalam spreadsheet, misalnya Copilot di
Microsoft Excel, Gemini di Google Sheets, dan integrasi ChatGPT untuk Excel. Fitur, nama menu, dan
ketersediaannya berubah cepat dan sebagian berbayar, jadi ebook ini tidak membahas tombol per tombol.

Prinsip di bab ini tetap berlaku:

AI yang bisa "melihat" file memang tidak perlu ditebak letak datanya, tapi kasus pinggir tetap harus
Anda uji.
Hasil yang ditulis AI ke sel tetap rumus atau angka biasa. Periksa apakah ia menulis rumus (bisa dicek dan
ikut berubah) atau angka mati (tidak ikut berubah saat data berubah).
Aturan privasi tetap berlaku: data yang dibaca AI di file tetap dikirim untuk diproses.

2.9  Ringkasan dan prompt siap pakai
AI menebak semua yang tidak Anda sebutkan. Sebutkan letak data, contoh hasil, versi Excel, dan pemisah.

Lima kesalahan AI: pemisah, fungsi versi baru, alamat sel, kasus pinggir, fungsi karangan.
Uji dengan 6–8 baris data uji termasuk sel kosong dan nilai batas. Cocokkan satu baris secara manual.

Saat meminta perbaikan, sertakan rumus, pesan error, dan isi sel yang dirujuk.
Jangan menempel data pribadi asli. Data contoh sudah cukup.

Kerangka prompt (salin dan isi bagian dalam kurung siku):

Saya [pekerjaan Anda] dan memakai Excel [versi] dengan pemisah rumus [titik koma/koma]. Di sheet "
[nama sheet]", [data apa] ada di kolom [huruf] mulai baris [angka]. Contoh isinya: [contoh fiktif]. Saya
ingin di kolom [huruf] tampil [hasil yang diinginkan], contohnya: [contoh hasil]. Kalau [kondisi khusus,
mis. sel kosong], hasilnya [apa]. Berikan rumus untuk baris [angka] yang bisa disalin ke bawah, jelaskan
tiap bagiannya, lalu buatkan 6 baris data uji termasuk kasus pinggir.

B A B  9

Umur, Masa Kerja, dan Tanggal Penting

Bu Rina di bagian HRD menerima pesan dari manajer pada hari Kamis sore: "Tolong kirim data karyawan per
akhir Oktober: umur, masa kerja, siapa yang pensiun tahun depan, dan kontrak siapa saja yang habis bulan
depan. Besok pagi, ya."

Ada 80 karyawan. Kalau dihitung manual dengan kalender, butuh berjam-jam dan rawan salah. Dengan
rumus yang tepat, cukup sepuluh menit. Bulan depan, cukup ganti satu sel tanggal.

Bab ini membangun tabel seperti itu dari nol. Hasil akhirnya ada di template 07-Data-Karyawan.xlsx .

Gambar 1. Tabel data karyawan: kolom kuning diisi, kolom abu-abu dihitung otomatis

9.1  Tanggal di Excel adalah angka
Sebelum menulis rumus tanggal apa pun, pahami satu hal: bagi Excel, tanggal adalah angka. Angka 1 berarti
1 Januari 1900, angka 2 berarti 2 Januari 1900, dan seterusnya. Tanggal 31 Oktober 2026 tersimpan sebagai
angka 46326. Yang Anda lihat sebagai 31/10/2026  hanyalah format tampilan.

Karena itu:

Tanggal bisa dijumlah dan dikurangi. =B5-A5  menghasilkan selisih hari.

Tanggal + 30 = tiga puluh hari kemudian.
Kalau sel tanggal tiba-tiba tampil sebagai 46326 , datanya tidak rusak. Formatnya saja berubah menjadi
General. Kembalikan dengan Ctrl+1 → Date.

Tanggal asli atau teks?

Masalah tanggal yang paling sering terjadi adalah tanggal yang tersimpan sebagai teks, biasanya hasil salin
dari sistem absensi, formulir online, atau file CSV. Ciri-cirinya:

isinya rata kiri, padahal tanggal asli rata kanan;

saat formatnya diubah ke General, tampilannya tidak berubah menjadi angka;
=ISNUMBER(F5)  menghasilkan FALSE .

Semua rumus di bab ini butuh tanggal asli. Kalau data Anda berupa teks, bersihkan dulu dengan cara di Bab 5
atau dengan template 02-Pembersih-Data.xlsx .

Jebakan. Mengetik 03/04/2026  di komputer berpengaturan Indonesia menghasilkan 3 April. Di
komputer berpengaturan Amerika, hasilnya 4 Maret. Kalau file dikerjakan bergantian, ketik tanggal
dengan nama bulan ( 3 Apr 2026 ) atau pakai rumus =DATE(2026;4;3) . Keduanya tidak ambigu.

9.2  Satu tanggal acuan untuk semua rumus
Umur dan masa kerja selalu dihitung sampai suatu tanggal. Ada dua pilihan:

TODAY() : selalu hari ini. Cocok untuk tampilan harian, misalnya pengingat kontrak.
Tanggal tetap, misalnya 31/10/2026. Cocok untuk laporan per akhir bulan, karena angkanya tidak
berubah saat file dibuka minggu depan.

Jangan menulis TODAY()  di setiap rumus. Letakkan tanggal acuan di satu sel di sheet terpisah, lalu rujuk sel
itu dari semua rumus. Di template 07, sel itu adalah Pengaturan!B4 , yang awalnya berisi =TODAY()  dan bisa
Anda timpa dengan tanggal tertentu.

Pengaturan!B4   →   =TODAY()      atau ketik   31/10/2026

Di semua rumus di bab ini, tanggal acuan ditulis Pengaturan!$B$4 , dikunci dengan tanda dolar agar aman
disalin ke bawah (Bab 1.3).

9.3  Umur dalam tahun penuh
Data karyawan di sheet Karyawan : tanggal lahir di kolom E, tanggal masuk di kolom F, data mulai baris 5.
Umur di kolom I:

=IF(E5="";"";DATEDIF(E5;Pengaturan!$B$4;"Y"))

DATEDIF(awal; akhir; "Y")  menghitung tahun penuh di antara dua tanggal. Orang yang besok
berulang tahun ke-36 masih dihitung 35.
IF(E5="";"";...)  membuat sel tetap kosong selama tanggal lahir belum diisi. Tanpa ini, baris kosong

akan menampilkan umur 126 tahun (dihitung dari "tanggal 0" di tahun 1900).

DATEDIF tidak muncul di daftar saran. Saat Anda mengetik =DAT , Excel menyarankan DATE  dan
DATEVALUE , tapi tidak DATEDIF . Fungsi ini disediakan untuk kompatibilitas dengan program

spreadsheet lama (Lotus 1-2-3), sehingga tidak ditampilkan di daftar saran. Fungsinya tetap bekerja di
semua versi Excel dan juga di Google Sheets. Ketik saja lengkap.

Kenapa tidak dibagi 365 saja?

Rumus yang sering beredar adalah =(Pengaturan!$B$4-E5)/365  lalu dibulatkan ke bawah. Rumus ini salah
di sekitar hari ulang tahun karena tahun kabisat:

Tabel 8. Rumus "bagi 365" vs DATEDIF, tanggal acuan 31/10/2026

Tanggal lahir Umur
sebenarnya

INT(selisih/365) INT(selisih/365,25) DATEDIF(...;"Y")

01/11/1990 (ulang tahun
besok)

35 36 ✗ 35 35

31/10/1992 (ulang tahun hari
ini)

34 34 33 ✗ 34

10/11/1990 35 35 35 35

Setiap rumus "bagi" punya tanggal-tanggal tertentu yang salah. DATEDIF  menghitung dengan kalender,
sehingga selalu tepat.

9.4  Umur dan masa kerja dalam tahun dan bulan
Untuk tampilan seperti 8 th 7 bln , gabungkan dua DATEDIF :

=IF(F5="";"";IF(F5>Pengaturan!$B$4;"Belum masuk";DATEDIF(F5;Pengaturan!$B$4;"Y")&" th

"&DATEDIF(F5;Pengaturan!$B$4;"YM")&" bln"))

Bedah per bagian:

Tabel 9. Bagian-bagian rumus masa kerja

Bagian Fungsinya

IF(F5="";"";...) Baris kosong tetap kosong

IF(F5>Pengaturan!$B$4;"Belum

masuk";...)

Karyawan yang tanggal masuknya setelah tanggal acuan. Tanpa ini,
DATEDIF  menghasilkan #NUM!

DATEDIF(...;"Y") Tahun penuh

DATEDIF(...;"YM") Sisa bulan setelah tahun penuh dibuang (0–11)

&" th "& Menggabungkan angka dengan teks

Satuan lain yang bisa dipakai DATEDIF :

Tabel 10. Satuan DATEDIF

Satuan Hasil Contoh (15/03/2018 s.d. 31/10/2026)

"Y" Tahun penuh 8

"M" Bulan penuh total 103

"D" Hari total 3152

"YM" Sisa bulan setelah tahun penuh 7

"MD" Sisa hari setelah bulan penuh 16

Jebakan. Microsoft sendiri memperingatkan bahwa satuan "MD"  bisa memberi hasil keliru pada
kombinasi tanggal tertentu, terutama di sekitar akhir bulan. Untuk dokumen resmi, cukup tampilkan
tahun dan bulan. Kalau memang perlu hari, hitung dari tanggal ulang bulan terakhir:
=Pengaturan!$B$4-EDATE(F5;DATEDIF(F5;Pengaturan!$B$4;"M")) .

Masa kerja sebagai angka desimal

Teks 8 th 7 bln  enak dibaca tapi tidak bisa dihitung. Untuk rata-rata masa kerja per divisi atau syarat
seperti "masa kerja minimal 2 tahun", butuh angka:

=IF(OR(F5="";F5>Pengaturan!$B$4);"";ROUND(YEARFRAC(F5;Pengaturan!$B$4;1);1))

YEARFRAC  menghitung pecahan tahun. Argumen ketiga 1  berarti memakai jumlah hari sebenarnya per
tahun, termasuk kabisat. Contoh di atas menghasilkan 8,6 tahun. Sheet Ringkasan  di template 07 memakai
kolom ini untuk rata-rata masa kerja per divisi dengan AVERAGEIF .

9.5  Tanggal pensiun
Tanggal pensiun adalah tanggal lahir ditambah usia pensiun dalam tahun. Penambahan dalam satuan bulan
paling aman dengan EDATE :

=IF(E5="";"";EDATE(E5;12*Pengaturan!$B$5))

EDATE(tanggal; jumlah_bulan)  maju sejumlah bulan dan menangani akhir bulan dengan benar. Lahir 31
Agustus, ditambah 6 bulan, jatuh di 28 atau 29 Februari, bukan 3 Maret. Usia pensiun disimpan di
Pengaturan!B5 , tidak ditulis langsung di rumus, karena aturannya berubah:

Tabel 11. Usia pensiun program Jaminan Pensiun (PP 45/2015)

Berlaku Usia pensiun

2019–2021 57 tahun

2022–2024 58 tahun

2025–2027 59 tahun

Mulai 2028 60 tahun, lalu naik 1 tahun setiap 3 tahun hingga 65

Usia pensiun di perusahaan Anda bisa berbeda karena diatur dalam perjanjian kerja, peraturan perusahaan,
atau PKB. Selalu ikuti aturan yang berlaku di tempat Anda.

Banyak perusahaan menetapkan pensiun di akhir bulan saat usia pensiun tercapai. Bungkus rumusnya
dengan EOMONTH :

=IF(E5="";"";EOMONTH(EDATE(E5;12*Pengaturan!$B$5);0))

Siapa yang pensiun dalam 12 bulan ke depan?

Di sheet Ringkasan , jumlah karyawan yang pensiun dalam setahun dihitung dengan SUMPRODUCT , karena
ada dua syarat tanggal sekaligus:

=SUMPRODUCT((Karyawan!$M$5:$M$304<>"")*(Karyawan!$M$5:$M$304>=Pengaturan!$B$4)*

(Karyawan!$M$5:$M$304<=EDATE(Pengaturan!$B$4;12)))

Setiap bagian dalam kurung menghasilkan 1 (benar) atau 0 (salah) untuk tiap baris. Ketiganya dikalikan,
sehingga hanya baris yang memenuhi semua syarat bernilai 1, lalu dijumlah.

Kenapa tidak COUNTIFS ? Bisa juga: COUNTIFS(M5:M304;">="&B4;M5:M304;"<="&EDATE(B4;12))  memberi hasil
yang sama. SUMPRODUCT  ditampilkan di sini karena polanya bisa dipakai untuk syarat yang tidak bisa ditulis di
COUNTIFS , misalnya syarat atas bulan dari suatu tanggal.

9.6  Kontrak PKWT yang akan berakhir
Untuk karyawan kontrak, kolom H berisi tanggal akhir kontrak. Sisa hari kontrak:

=IF(OR(G5<>"PKWT";H5="");"";H5-Pengaturan!$B$4)

Karena tanggal adalah angka, pengurangan langsung menghasilkan jumlah hari. Lalu status kontraknya:

=IF(N5="";"";IF(N5<0;"Sudah berakhir";IF(N5<=Pengaturan!$B$6;"Segera berakhir";"Aktif")))

Pengaturan!B6  berisi batas peringatan, misalnya 30 hari. Supaya terlihat, beri warna dengan Conditional
Formatting:

1. Pilih kolom status (mis. O5:O304 ).

2. Home → Conditional Formatting → New Rule → Use a formula to determine which cells to format.
3. Isi rumus =$O5="Segera berakhir" , pilih warna kuning. Buat aturan kedua =$O5="Sudah berakhir"

dengan warna merah.

Tips. Tanda dolar di $O5  mengunci kolom tapi tidak barisnya. Dengan begitu, aturan yang sama bisa
diterapkan ke seluruh baris tabel, misalnya A5:P304 , sehingga satu baris penuh berwarna, bukan
hanya sel statusnya.

9.7  Ulang tahun kerja bulan ini
Banyak kantor memberi ucapan atau penghargaan masa kerja. Untuk menandai karyawan yang ulang tahun
kerjanya jatuh di bulan tanggal acuan:

=IF(F5="";"";IF(AND(MONTH(F5)=MONTH(Pengaturan!$B$4);F5<Pengaturan!$B$4);"Ya -

"&DATEDIF(F5;EOMONTH(Pengaturan!$B$4;0);"Y")&" tahun";""))

MONTH(F5)=MONTH(acuan)  mencocokkan bulannya saja.
F5<acuan  mencegah karyawan yang baru masuk bulan ini ikut ditandai.

DATEDIF  sampai akhir bulan ( EOMONTH ) memastikan angkanya sudah termasuk ulang tahun kerja yang
jatuh di akhir bulan. Hasilnya misalnya Ya - 10 tahun .

Pola yang sama berlaku untuk ulang tahun kelahiran: ganti F5 dengan kolom tanggal lahir.

9.8  Jebakan umum
Tabel 12. Gejala, penyebab, dan perbaikan

Gejala Penyebab Perbaikan

Umur 126 tahun Tanggal lahir kosong, dihitung dari tahun
1900

Tambahkan IF(E5="";"";...)

#VALUE! Tanggal tersimpan sebagai teks Bersihkan data (Bab 5, template 02)

#NUM! Tanggal awal lebih besar dari tanggal akhir Tambahkan pengecekan
IF(F5>acuan;...)

Umur tampil 04/02/1900 Sel hasil berformat tanggal Ubah format sel ke General atau Number

Hasil berubah tiap hari Memakai TODAY()  di laporan bulanan Pakai tanggal acuan tetap di satu sel

Tanggal 3 April jadi 4
Maret

Pengaturan tanggal berbeda antar komputer Ketik dengan nama bulan atau DATE()

9.9  Excel lama dan Google Sheets
Semua fungsi di bab ini, yaitu DATEDIF , YEARFRAC , EDATE , EOMONTH , MONTH , dan SUMPRODUCT , tersedia di
Excel 2007 ke atas dan di Google Sheets dengan nama dan cara pakai yang sama. Tidak perlu versi pengganti.

9.10  Prompt AI untuk bab ini

Saya staf HRD, memakai Excel [versi] dengan pemisah titik koma. Di sheet "Karyawan", tanggal lahir di
kolom E dan tanggal masuk di kolom F, mulai baris 5. Tanggal acuan di Pengaturan!B4. Buatkan rumus
untuk [pilih: umur dalam tahun / masa kerja "X th Y bln" / tanggal pensiun usia 59 di akhir bulan / tanda
karyawan yang masa kerjanya tepat 5, 10, atau 15 tahun di bulan acuan]. Kalau tanggal kosong, hasil
kosong. Jelaskan tiap bagian dan beri 6 baris data uji termasuk tanggal 29 Februari dan tanggal yang
jatuh tepat di tanggal acuan.

9.11  Template: 07-Data-Karyawan.xlsx
Tabel 13. Isi template 07

Sheet Isi

Petunjuk Cara pakai dan arti warna sel

Pengaturan Tanggal acuan (B4), usia pensiun (B5), batas peringatan kontrak (B6)

Karyawan Data hingga 300 karyawan; umur, masa kerja, pensiun, sisa kontrak, ulang tahun kerja

Ringkasan Jumlah per divisi dan jenis kelamin, rata-rata umur dan masa kerja, jumlah kontrak berakhir, pensiun dalam 12
bulan

Cara memakainya untuk laporan bulanan Bu Rina:

1. Isi data karyawan di kolom kuning A–H.

2. Di sheet Pengaturan, ganti B4 dengan tanggal akhir bulan laporan, misalnya 31/10/2026 .
3. Saring kolom "Status Kontrak" untuk melihat yang segera berakhir, dan kolom "Tgl Pensiun" untuk yang

pensiun dalam setahun.
4. Bulan depan, cukup ganti B4 ke 30/11/2026 . Semua angka ikut menyesuaikan.

Sepuluh menit, dan Bu Rina bisa pulang tepat waktu.

Isi Ebook Lengkap

Anda baru membaca contoh gratis. Ebook lengkap berisi 23 bab, 4 lampiran, dan 24 template Excel siap
pakai. Bab yang dicetak tebal ada di contoh ini.

1. Sebelum Menulis Rumus

2. Menyuruh AI Membuat Rumus (dan
Memeriksanya)

3. 20 Rumus yang Dipakai Setiap Hari
4. Mencari Data: XLOOKUP dan Penggantinya
5. Membersihkan Data Berantakan

6. Rekap & Laporan Bulanan dengan PivotTable
7. Inventaris dan Stok Barang
8. Surat Massal dan Penomoran Dokumen
9. Umur, Masa Kerja, dan Tanggal Penting

10. Rekap Absensi dan Lembur
11. Slip Gaji Sederhana (BPJS & PPh 21 TER)

12. Cuti dan Penilaian Kinerja

13. Terbilang Rupiah Tanpa Add-in
14. Kuitansi, Invoice, dan Nota
15. Kas Kecil, Piutang, dan Laba Rugi
16. Rekonsiliasi Bank
17. Rekap Nilai dan Rapor

18. Absensi Siswa
19. Dashboard Satu Halaman
20. Data Validation, Proteksi, dan Formulir Input
21. Google Sheets untuk Kerja Tim

22. Otomatisasi Ringan
23. Latihan Soal Tes Excel Admin (30 soal, dicek

otomatis)

Dapatkan ebook lengkap + 24 template di ebook.aresstudio.id/excel-kantor-ai
File kerja untuk daftar harga, stok, nomor surat, data karyawan, absensi, payroll, cuti, KPI, terbilang,
invoice, kas kecil, umur piutang, laba rugi, rekonsiliasi bank, rapor, absensi siswa, dashboard, formulir
aman, dan latihan tes.


